
* Program.: RPRTPRO.PRG 
* Author..: Stephen Kurasch 
* Date....: June 10, 1985 
* Notice..: Copyright 1985, Ashton-Tate 
* Notes...: Procedure file for all reports for the 
* ........: checkbook management system 
* ........: Called from REPORTS.PRG 
 
PROCEDURE Reporta 
   * erase menu title to make room for new menu title 
   @ 3,30 SAY SPACE(20) 
 
   * set up sub-menu 
   DO CASE 
      CASE choice = "1" 
         option = "Checks" 
         col = 31 
         ub = 5 
         @ 12,0 CLEAR 
         @ 12, 0 SAY dvert 
         @ 12,79 SAY dvert 
         @ 13, 0 SAY CHR(200)+dhoriz+CHR(188) 
      CASE choice = "2" 
         option = "Bank Deposits" 
         col = 28 
         ub = 4 
      CASE choice = "3" 
         option = "Individual Deposits" 
         col = 25 
         ub = 3 
         @ 11,0 CLEAR 
         @ 11, 0 SAY CHR(200)+dhoriz+CHR(188) 
      CASE choice = "4" 
         option = "Cash Withdrawal/Automatic Teller" 
         col = 18 
         ub = 4 
   ENDCASE 
 
   * display new sub-menu and erase first choice selection 
   @ 3,col SAY UPPER(option) + " SUB-MENU" 
   @ 6+VAL(choice),24 SAY SPACE(30) 
 
   * special condition for option 
   IF choice = "4" 
      option = "Cash Withdrawals" 
   ENDIF 
 
   @ 6,20 SAY "0 - Return to REPORTS MENU" 
   @ 7,20 SAY "1 - List all "+option 
   @ 8,20 SAY "2 - List all "+option+" between two dates" 
   @ 9,20 SAY "3 - List all "+option+" over a certain amount" 
 
   * set up files and commands for reports 
   DO CASE 



      CASE choice = "1" 
         @ 10,20 SAY "4 - List all taxable checks" 
         @ 11,20 SAY "5 - List certain taxable checks" 
         USE Checks 
         command1 = "   CHKNO    DATE         PAY TO THE ORDER OF"+; 
                    "          AMOUNT" 
         command2 = "   -----  --------  ------------------------------  
"+; 
                    "----------" 
         command = "'    '+STR(Chkno,4,0)+'  '+DTOC(Date)+'  '+Payto+'  
'+; 
                   STR(Amt,10,2)" 
      CASE choice = "2" 
         @ 10,20 SAY "4 - List all deposits in transit" 
         USE Bank 
         command1 = "     DATE      AMOUNT" 
         command2 = "   --------  ----------" 
         command = "'   '+DTOC(Date)+'  '+STR(Amt,10,2)" 
      CASE choice = "3" 
         USE Deposits 
         command1 = "     DATE            PAIDFROM             AMOUNT" 
         command2 = "   --------  -------------------------  ----------" 
         command = "'   '+DTOC(Date)+'  '+Paidfrom+'  '+STR(Amt,10,2)" 
      CASE choice = "4" 
         @ 10,20 SAY "4 - List all Cash Withdrawals in transit" 
         USE Bank 
         command1 = "     DATE      AMOUNT" 
         command2 = "   --------  ----------" 
         command = "'   '+DTOC(Date)+'  '+STR(-Amt,10,2)" 
   ENDCASE 
 
   * select sub-menu choice2 
   choice2 = 6 
   DO WHILE choice2 < 0 .OR. choice2 > ub 
      mchoice2 = " " 
      @ 8+ub,33 SAY " select : : " 
      @ 8+ub,42 GET mchoice2 
      READ 
      choice2 = VAL(mchoice2) 
   ENDDO 
 
   * display asterisk by second choice 
   @ 6+choice2,18 SAY "*" 
 
   * set up exit condition for sub-menu - return to reports menu 
   IF choice2 = 0 
      RETURN 
   ENDIF 
 
   * erase other sub-menu options from the screen 
   cntr = 0 
   DO WHILE cntr <= ub 
      IF cntr <> choice2 
         @ 6+cntr,20 SAY SPACE(55) 



      ENDIF 
      cntr = cntr + 1 
   ENDDO 
 
   * set up new row for input messages 
   newrow = ub + 10 
 
   pr = " " 
   repo1 = "Reporta" + mchoice2 
   DO &repo1 
 
   ? 
   ? 
   IF UPPER(pr) = "P" 
      SET PRINT OFF 
   ENDIF 
   WAIT SPACE(15)+"Press any key to return to the REPORTS MENU " 
   CLOSE DATABASES 
 
PROCEDURE Reporta1 
 
*--------------------------------------------------------------* 
*                                                              * 
*    Programming Note:  The next lines are discussed in        * 
*    Chapter 16 of "Programming in dBASE III PLUS"             * 
*                                                              * 
*--------------------------------------------------------------* 
 
   * ask for output to printer or screen 
   DO Printer WITH newrow,pr 
 
   ? 
   ? SPACE(27)+UPPER(option) 
   ? command1 
   ? command2 
   DO WHILE .NOT. EOF() 
      DO CASE 
         CASE choice = "2" 
            IF Amt > 0.00 
               ? &command 
            ENDIF 
         CASE choice = "4" 
            IF Amt < 0.00 
               ? &command 
            ENDIF 
         OTHERWISE 
            ? &command 
      ENDCASE 
      SKIP 
   ENDDO 
 
PROCEDURE Reporta2 
   * initialize memory variables 
   STORE CTOD("  /  /  ") TO begin,end 



   amount = 0.00 
 
   DO WHILE .T. 
 
      * input valid starting date 
      @ newrow,23 SAY "Enter starting date" GET begin 
      READ 
 
      * input valid ending date 
      @ newrow+2,23 SAY "Enter ending date  " GET end 
      READ 
 
      * check to make sure starting date comes before ending date 
      IF begin <= end 
         EXIT 
      ENDIF 
 
      @ 20,13 SAY "Ending date occurs before starting date - please 
reenter" 
 
   ENDDO 
   @ 20,0 
 
   * output to printer or screen? 
   DO Printer WITH 20,pr 
 
   ? 
   ? SPACE(15)+UPPER(option)+" BETWEEN "+DTOC(begin)+" AND "+DTOC(end) 
   ? command1 
   ? command2 
 
   DO WHILE .NOT. EOF() 
      IF Date >= begin .AND. Date <= end 
         DO CASE 
            CASE choice = "2" 
               IF Amt > 0.00 
                  ? &command 
                  amount = amount + Amt 
               ENDIF 
            CASE choice = "4" 
               IF Amt < 0.00 
                  ? &command 
                  amount = amount - Amt 
               ENDIF 
            OTHERWISE 
               ? &command 
               amount = amount + Amt 
         ENDCASE 
      ENDIF 
      SKIP 
   ENDDO 
 
   first = IIF(choice="1",SPACE(43),IIF(choice="3",SPACE(31),SPACE(4))) 
 



   ? 
   ? first+"TOTAL   $"+STR(amount,10,2) 
 
PROCEDURE Reporta3 
   DO WHILE .T. 
 
      * input valid amount 
      amount = -1.00 
      DO WHILE amount < 0.00 
         amount = 0.00 
         @ newrow,20 SAY "Enter in amount" GET amount 
         READ 
      ENDDO 
 
      @ newrow+2,10 SAY "All "+option+" listed will be over $"+; 
                        LTRIM(STR(amount,10,2)) 
 
      * valid amount entered - check for correctness 
      answer = " " 
      DO WHILE .NOT. answer$"YyNn" 
         answer = " " 
         @ newrow+3,20 SAY "Is this correct? (Y/N)" GET answer 
         READ 
      ENDDO 
 
      IF UPPER(answer) ="Y" 
         EXIT 
      ENDIF 
      @ newrow+2,0 CLEAR 
 
   ENDDO 
 
   * output to printer or screen? 
   DO Printer WITH 20,pr 
 
   GO TOP 
   ? 
   ? SPACE(13)+UPPER(option)+" OVER $"+LTRIM(STR(amount,10,2)) 
   ? command1 
   ? command2 
 
   DO WHILE .NOT. EOF() 
      IF choice = "4" 
         IF -Amt >= amount 
            ? &command 
         ENDIF 
      ELSE 
         IF Amt >= amount 
            ? &command 
         ENDIF 
      ENDIF 
      SKIP 
   ENDDO 
 



PROCEDURE Reporta4 
   IF choice = "1" 
      DO Report14 
   ELSE 
      DO Report24 
   ENDIF 
 
PROCEDURE Report14 
   SELECT 2 
   USE Tax 
 
   * output to printer or screen? 
   DO Printer WITH 21,pr 
 
   ? 
   ? SPACE(20)+"TAXABLE CHECKS" 
   ? 
 
   DO WHILE .NOT. EOF() 
      ? SPACE(20)+Title 
      ? "   CHKNO    DATE         PAY TO THE ORDER OF          AMOUNT" 
      ? "   -----  --------  ------------------------------  ----------" 
 
      amount = 0.00 
 
      SELECT 1 
      GO TOP 
      DO WHILE .NOT. EOF() 
         IF Tax = B->Code 
            ? "    "+STR(Chkno,4,0)+"  "+DTOC(Date)+"  "+Payto+"  "+; 
              STR(Amt,10,2) 
            amount = amount + Amt 
         ENDIF 
         SKIP 
      ENDDO 
 
      ? 
      ? SPACE(42)+"TOTAL    $" + STR(amount,10,2) 
      ? 
      ? 
 
      SELECT 2 
      SKIP 
   ENDDO 
 
PROCEDURE Report24 
   IF choice = "2" 
      header = "          LIST ALL DEPOSITS IN TRANSIT" 
      compare = "Amt > 0" 
      newcol = 30 
      statement = "No deposits in transit" 
   ELSE 
      header = "LIST ALL CASH WITHDRAWAL/AUTOMATIC TELLER TRANSACTIONS 
"+; 



               "IN TRANSIT" 
      compare = "Amt < 0" 
      newcol = 5 
      statement = "No Cash Withdrawal/Automatic Teller transactions in 
transit" 
   ENDIF 
 
   LOCATE FOR .NOT. Clear .AND. &compare 
 
   IF EOF() 
      @ newrow,newcol SAY statement 
      RETURN 
   ENDIF 
 
   * output to printer or screen? 
   DO Printer WITH 20,pr 
 
   ? 
   ? header 
   ? command1 
   ? command2 
 
   amount = 0.00 
   DO WHILE .NOT. EOF() 
      IF .NOT. Clear .AND. &compare 
         ? &command 
         amount=IIF(choice="2",amount+Amt,amount-Amt) 
      ENDIF 
      SKIP 
   ENDDO 
 
   ? 
   ? "    TOTAL   $"+STR(amount,10,2) 
 
PROCEDURE Reporta5 
   SELECT 2 
   USE Tax 
 
   * display all tax codes 
   mrow = 15 
   mcol = 1 
   DO WHILE .NOT. EOF() 
      @ mrow,mcol SAY "("+STR(Code,1)+") "+Title 
      mrow=mrow+1 
      IF mrow=18 
         mrow=15 
         mcol=mcol+27 
      ENDIF 
      SKIP 
   ENDDO 
 
   * select specific tax deductible check 
   mtax = 0 
   @ 19,6 SAY "Enter specific tax expense (or 0 to return "+; 



              "to the REPORTS MENU)" GET mtax PICTURE "9" 
   READ 
 
   * test for exit condition 
   IF mtax = 0 
      RETURN 
   ENDIF 
 
   * set up tax expense option to be processed 
   GO mtax 
 
   * output to printer or screen? 
   DO Printer WITH 21,pr 
 
   ? 
   ? SPACE(25) + Title 
   ? 
   ? "   CHKNO    DATE         PAY TO THE ORDER OF          AMOUNT" 
   ? "   -----  --------  ------------------------------  ----------" 
 
   amount = 0.00 
   SELECT 1 
   GO TOP 
 
   DO WHILE .NOT. EOF() 
      IF Tax = mtax 
         ? "    "+STR(Chkno,4,0)+"  "+DTOC(Date)+"  "+Payto+"  
"+STR(Amt,10,2) 
         amount = amount + Amt 
      ENDIF 
      SKIP 
   ENDDO 
 
   ? 
   ? SPACE(42)+"TOTAL    $"+STR(amount,10,2) 
 
*--------------------------------------------------------------* 
*                                                              * 
*    Programming Note:  The following section is discussed in  * 
*    Chapters 12 and 16 of "Programming in dBASE III PLUS"     * 
*                                                              * 
*--------------------------------------------------------------* 
 
PROCEDURE Printer 
   PARAMETERS row,pr 
   @ row,6 SAY "Do you want the output sent to the " +; 
               "printer or the screen? (P/S)" 
   pr = " " 
   DO WHILE .NOT. pr$"PpSs" 
      pr = " " 
      @ row,70 GET pr 
      READ 
   ENDDO 
   CLEAR 



   IF UPPER(pr) = "P" 
      SET PRINT ON 
   ENDIF 
* Eof: 

